

# 1 Centralne twierdzenia graniczne symulacje

## 1.1 Tworzenie rozkładu empirycznego zmiennych losowych

Aby znaleźć empiryczny rozkład zmiennej losowej musimy wygenerować jej realizacje. Program robi to w następujący sposób. Generuje realizacje niezależnych zmiennych losowych o rozkładzie jednostajnym  $u_1, u_2, \dots, u_n$  a następnie korzysta z dystrybuanty odwrotnej. Zatem generowanie dotyczy sytuacji kiedy mamy dystrybuantę odwrotną. Program robi to automatycznie. Pokażemy to na przykładzie rozkładu Beta. Rozkład ten ma dwa parametry  $a, b$ . Zrobimy to też dla rozkładu wykładniczego. W pakiecie SAS 9, który wykorzystujemy rozkład wykładniczy ma domyślnie parametr  $\lambda = 1$ .

```
data reuters0;
    do i = 1 to 2000;
        z = rand('Beta',1,9);
        output;
    end;
run;
```

## 1.2 Wizualizacja rozkładu empirycznego czyli estymacja gęstości

Aby wykorzystać odpowiednią procedurę SAS tworzenia estymatora gęstości ustawia się tzw. parametr okna `bwm` oraz granice, górną i dolną gdzie leży ten rozkład. Zaobserwujemy, że zmieniając parametr `bwm` w granicach od 0.1 – 2 następuje wygładzanie. Parametr `bwm` zbyt duży wygląda absurdalnie.

```
ods graphics on;
proc kde data=reuters1; univar z/ gridl=0 gridu=1 bwm=1; run;
ods graphics off;
```

### 1.3 Centralne twierdzenie graniczne. Wartość oczekiwana i wariancja znane

Rozważamy tutaj sytuację dwóch rozkładów. Wykładniczego i jednostajnego. Interesuje nas rozkład empiryczny zmiennej losowej

$$S_n = \sum_{j=1}^n X_j,$$

gdzie  $X_1, \dots, X_n$  jest ciągiem zmiennych losowych o jednakowym rozkładzie. Zmienną  $S_n$  "przeskalujemy". Proces ten nazywany jest również "procesem standaryzacji".

```
data reuters1; x=0; z=0;

do i = 1 to 2000;
  x=0;
  do k=1 to 2;
    x = rand('Exponential')+x;
  end;
  z=(x-2)/sqrt(2);
  output;
end;
run;
```

Wartość oczekiwana i wariancja są odpowiednio równe:

dla rozkładu 'Exponential' wykładniczego  $VarX = 1, EX = 1$

dla rozkładu 'Uniform' jednostajnego  $VarX = 1/12, EX = 1/2$ .

Zatem normowanie  $z$  dla  $n = 2$

$$z = \frac{x - n * EX}{\sqrt{nVarX}}$$

jest konsekwencją wzoru

$$\frac{S_n - ES_n}{\sqrt{VarS_n}}.$$

Wynika to stąd, że jeśli mamy sumę niezależnych zmiennych losowych  $X_1, \dots, X_n$  o jednakowym rozkładzie zgodnym z  $X$ , oraz

$$S_n = X_1 + \dots + X_n,$$

to

$$ES_n = EX_1 + \dots + EX_n = nEX$$

oraz

$$VarS_n = VarX_1 + \dots + VarX_n = nVarX.$$

Pamiętamy, że przy  $n \rightarrow \infty$

$$\frac{S_n - ES_n}{\sqrt{VarS_n}} \rightarrow N(0, 1).$$

co obserwujemy zwiększając  $n$ . Rozpoczynamy od  $n = 2$

## 1.4 Centralne twierdzenie graniczne. Wartość oczekiwana i wariancja nieznane

Będziemy obserwować wizualizację uogólnienia centralnego twierdzenia granicznego, w sytuacji gdyby wartość oczekiwana i wariancja nie były znane. My będziemy korzystać ze znanych rozkładów: Beta, Weibull, Gamma. Dla SAS Beta ma dwa parametry, Gamma – jeden parametr, Weibull – dwa parametry. Zamiast  $ES_n$  oraz wariancji  $VarS_n$  wstawiamy ich estymatory. Przy pewnych założeniach twierdzenia – mocne(słabe) prawo wielkich liczb – estymatory te są zbieżne prawie wszędzie, (prawie napewno p.n.) (punktowo).

Niech dany będzie ciąg niezależnych zmiennych losowych

$$\xi_1, \dots, \xi_n, \dots$$

o jednakowym rozkładzie zgodnym z  $\xi$ . Wówczas

**Mocne prawo wielkich liczb**

$$\frac{1}{n} \sum_{j=1}^n \xi_j \rightarrow E\xi \quad p.n.$$

**Słabe prawo wielkich liczb**

$$\frac{1}{n} \sum_{j=1}^n \xi_j \rightarrow E\xi \quad wg. \text{prawdop.}$$

Aby wykorzystać te twierdzenia tworzymy niezależne kopie  $S_n$ , czyli

$$\xi_1 = S_{n1}, \dots, \xi_k = S_{nk}, \dots$$

i stosuje powyższe twierdzenie do oszacowania  $ES_n$ . Analogicznie dla  $VarS_n$ . Ponieważ

$$Var\xi = E\xi^2 - (E\xi)^2,$$

zatem wykorzystujemy (przy pewnych założeniach) fakt, że

$$\frac{1}{n} \sum_{j=1}^n \xi_j^2 \rightarrow E\xi^2 \quad p.n. \quad lub \quad wg. \quad prawdop.$$

To wystarczy aby dostać centralne twierdzenie graniczne.

```
data reuters2; array r[2000] r1-r2000;
```

```
do i = 1 to 2000;
  x=0;

  do k=1 to 2;
    x = Rand('Beta',1,9)+x;
  end;
  r[i]=x;
end;
```

\* Wyznaczamy średnią oraz odchylenie standardowe empiryczne\*

```
  suma=0;
  sumakw=0;
  do i=1 to 2000;
    suma=r[i]+suma;
    sumakw=r[i]**2+sumakw;
  end;

  sr=suma/2000;
  var=sumakw/2000-(sr)**2;
  odch=sqrt(var);

  do i=1 to 2000;
    z1=(r[i]-sr)/odch;
  output;
  end;
run;
```

## 1.5 Centralne twierdzenie graniczne. Wartość oczekiwana i wariancja nieznane, suma zależy od zmiennej Poissona

Rozważymy teraz sytuację zmiennej losowej

$$S_N = \sum_{j=1}^N X_j,$$

gdzie  $X_1, \dots, X_n$  jest ciągiem zmiennych losowych o jednakowym rozkładzie. Ponadto dana jest zmienna losowa  $N$  o rozkładzie Poissona z parametrem  $\lambda$ . Parametr  $\lambda$  jest równy wartości oczekiwanej, czyli  $EN = \lambda$ . Zmienna losowa  $N$  jest niezależna od ciągu  $X_1, \dots, X_n$ . Można udowodnić i zaobserwować, że jeśli  $\lambda \rightarrow \infty$ , to również zachodzi centralne twierdzenie graniczne. Dla prostoty zapiszmy je przy znanych  $ES_n$  oraz  $VarS_n$ . Zatem jeśli  $\lambda \rightarrow \infty$ , to

$$\frac{S_N - ES_N}{\sqrt{VarS_N}} \rightarrow N(0, 1).$$

```
data reuters3;
```

```
array r[2000] r1-r2000;
```

```
do i = 1 to 2000;
  poss=Rand('Poisson',3);
  x=0;

  do k=1 to poss;
    x = Rand('Beta',1,9)+x;
  end;
  r[i]=x;
end;
```

\* Beta dwa parametry Gamma jeden parametr Weibull dwa parametry. Wyznaczamy średnią oraz odchylenie standardowe empiryczne\*

```
suma=0;
sumakw=0;
do i=1 to 2000;
  suma=r[i]+suma;
  sumakw=r[i]**2+sumakw;
end;
```

```

sr=suma/2000;
var=sumakw/2000-(sr)**2;
odch=sqrt(var);

do i=1 to 2000;
z1=(r[i]-sr)/odch;
output;
end;
run;

```

## 1.6 Centralne twierdzenie graniczne dla rozkładu Cauchy

Rozkład Cauchy dany jest przez gęstość dla  $x \in R$

$$f(x) = \frac{1}{\pi(1+x^2)}.$$

Przyjrzymy się bwm

```

data cauchy0;

do i = 1 to 20000;
z = rand('Cauchy');
output;
end;
run;

ods graphics on;

proc kde data=cauchy0; univar z/grid1=-8 gridu=8 bwm=0.05; run;

ods graphics off;

```

Jeśli zmienna  $X$  ma rozkład Cauchy, to  $EX$  nie istnieje. Mimo to jest sens rozpatrywanie

$$S_n = \sum_{j=1}^n X_j,$$

gdzie  $X_1, \dots, X_n$  jest ciągiem zmiennych losowych o jednakowym rozkładzie Cauchy  $X$ . Tutaj możemy skorzystać z teorii funkcji charakterystycznych. Funkcja charakterystyczna zmiennej losowej  $X$  jest zdefiniowana przez

$$\phi_X(t) = Ee^{itX}.$$

Funkcja charakterystyczna identyfikuje rozkład zmiennej losowej. Jeśli dane są dwie zmienne losowe  $X, Y$  niezależne, to zachodzi twierdzenie

$$\phi_{X+Y}(t) = \phi_X(t)\phi_Y(t)$$

dla wszystkich  $t \in R$ . Funkcja charakterystyczna dla rozkładu Cauchy wynosi

$$\phi_X(t) = e^{-|t|}.$$

Zatem jeśli  $X_1, \dots, X_n$  jest ciągiem zmiennych losowych o jednakowym rozkładzie Cauchy  $X$ , to

$$\phi_{X_1+\dots+X_n}(t) = (e^{-|t|})^n = e^{-|nt|} = \phi_X(nt) = \phi_{nX}(t).$$

Zatem jest to przeskalowany rozkład Cauchy czyli

$$\frac{1}{n}S_n = \frac{1}{n} \sum_{j=1}^n X_j \stackrel{d}{=} X.$$

Jest to tzw. rozkład stabilny z  $\alpha = 1$  obserwujemy to w programie

```
data cauchy;
```

```
n=20;
```

```
  do i = 1 to 2000;
```

```
    x=0;
```

```
    do k=1 to n;
```

```
      x = Rand('Cauchy')+x;
```

```
    end;
```

```
    z=x/n;
```

```
    output;
```

```
  end;
```


```
run;
```

```
ods graphics on;
```

```
proc kde data=cauchy; univar z/grid1=-7 gridu=7 bwm=0.01; run;
```

```
ods graphics off;
```

Otrzymany rozkład wygląda jak normalny.



Oczywiście taki nie jest. Kryteria statystyczne pozwalają zdecydować jaki mamy rozkład.

## 1.7 Teoria ryzyka-wycena przedsiębiorstwa

Zanim przejdziemy do wyceny przedsiębiorstwa metodą Samuelsona podamy dwa inne modele wyceny. Kluczowym narzędziem jest proces Poissona.

Proces  $N(t)$ ,  $t \geq 0$  o własnościach

1. Przyrosty niezależne, czyli dla dowolnych  $t_1 < t_2 < t_3 < t_4$ , zmienne losowe

$$N(t_2) - N(t_1), \quad N(t_4) - N(t_3)$$

są niezależne.

2. Proces startuje z zera  $P(N(0) = 0) = 1$ .

3. Przyrosty,  $N(t) - N(s)$  mają rozkład Poissona z parametrem  $\lambda(t - s)$ .

Proces to w istocie funkcja dwóch zmiennych

$$N : [0, \infty) \times \Omega \rightarrow R$$

spełniająca pewne warunki zgodności. Używamy przy tym pewnych skrótów  $N(t, \omega) = N(t)(\omega)$ . Warunki zgodności pozwalają przy ustalonym  $t$ , rozpatrywać  $N(t)$  jako zmienną losową o warunkach 1-3. Zgodnie z definicją jest to zmienna losowa o rozkładzie Poissona gdyż  $N(0) = 0$  prawie wszędzie.

W istocie to dopiero umiejętność (wyznaczenia) symulacji trajektorii, pozwala nam myśleć, że znamy proces. Zatem interesuje nas jak dla ustalonego  $\omega \in \Omega$  wyznaczyć funkcję  $N(t)$ ,  $t \geq 0$  (opuszczamy dla procesów  $\omega$ ). Wyznamy "wszystkie" trajektorie procesu Poissona. W tym celu posłużymy się inną definicją.

Niech  $T_1, \dots, T_n$  będzie ciągiem zmiennych losowych o jednakowym rozkładzie wykładniczym z parametrem  $\lambda$ . Wówczas proces Poissona możemy zdefiniować za pomocą formuły  $N(t) = 0$  gdy  $T_1 > t$  oraz jeśli  $T_1 \leq t$  wówczas

$$N(t) = \sup\{n : T_1 + T_2 + \dots + T_n \leq t\}.$$

Bardzo często interpretuje się  $N(t)$  jako ilość zgłoszeń do centrali telefonicznej, ilość roszczeń w twarzystwie ubezpieczeniowym, ilość kontaktów. Czas  $T_1$  to chwila pojawienia się pierwszego zgłoszenia roszczenia itd. Z definicji wynika, że proces w chwili  $T_1$  przyjmuje wartość 1, w chwili  $T_k$  przyjmuje wartość  $k$ . Proces zatem przeskakuje tylko o jeden poziom. Zatem w danej chwili rejestruje się tylko jedno roszczenie. Podać trajektorię procesu, tj. podać wszystkie możliwe scenariusze pojawienia się roszczenia. Parametr  $\lambda$  oznacza w tym przypadku oczekiwaną liczbę roszczeń w określonym okresie czasu (jeden rok).

Bardziej rozwinięty model to mieszany proces Poissona. To jest model w którym parametr  $\lambda$  jest sterowny niezależnie poprzez zmienną losową  $\Lambda$ . Wygląda to następująco, najpierw wybieramy  $\lambda$  wg. rozkładu zmiennej losowej  $\Lambda$ . Następnie przeprowadzamy całą analizę wybierając czasy pojawienia się roszczeń (najrozsądniej jest ustalić skończoną perspektywę czasową np.  $0 \leq t \leq 10$  lat. Czasy roszczeń symulujemy aż przekroczy w sumie 10 lat.

### 1.7.1 Cramer Lundberg

Model sprawności towarzystwa ubezpieczeniowego. To model w którym ilość roszczeń modeluje proces Poissona (ew. mieszany proces Poissona)

$$U(t) = u + ct - S(t),$$

gdzie  $S(t)$  jest procesem skumulowanych roszczeń do chwili  $t$  czyli

$$S(t) = \sum_{j=1}^{N(t)} X_j,$$

gdzie  $N(t)$  to liczba roszczeń do chwili  $t$  zaś  $X_j$  to wielkość  $j$  tego roszczenia. Ponadto  $u$  to kapitał początkowy, zaś  $c$  to intensywność wpłat. Modelowanie trajektorii tego procesu pokaże nam w ilu hipotetycznych przypadkach proces  $U(t) < 0$  dla  $0 \leq t \leq 10$ . To jest metoda odpowiedzi na pytanie jakie jest prawdopodobieństwo bankructwa w przedziale  $0 \leq t \leq 10$ , czyli

$$P(U(t) < 0, 0 \leq t \leq 10) = ?$$

### 1.7.2 Wycena małego przedsiębiorstwa

Ten model w pewnym sensie jest odwrotny to modelu Cramera Lundberga. Mianowicie w modelu tym zakłada się, że jest kapitał założycielski  $u$ ,  $c$  to stała intensywność kosztów stałych zaś  $X_j$  to zysk z pozyskanego kontraktu (czyli przychód pomniejszy o koszt zmienny). Zatem

$$U(t) = u - ct + S(t),$$

gdzie  $S(t)$  jest procesem skumulowanych zysków z kolejnych kontraktów do chwili  $t$ , czyli

$$S(t) = \sum_{j=1}^{N(t)} X_j.$$

### 1.7.3 Symulacje $S(t)$

Aby wykonać symulacje musimy użyć konkretnych rozkładów. Dochodzimy tutaj to problematyki estymacji parametrów modelu (rozkładu) lub estymacji rozkładów roszczeń (kontraktów).

## 2 Procesy Poissona i Wienera

### 2.1 Proces Poissona

Proces  $N(t)$ ,  $t \geq 0$  o własnościach

1. Przyrosty niezależne, czyli dla dowolnych  $t_1 < t_2 < t_3 < t_4$ , zmienne losowe

$$N(t_2) - N(t_1), \quad N(t_4) - N(t_3)$$

są niezależne.

2. Proces startuje z zera  $P(N(0) = 0) = 1$ .

3. Przyrosty,  $N(t) - N(s)$  mają rozkład Poissona z parametrem  $\lambda(t - s)$ .

```
data poisson;
```

```
  t=0;
```

```
  p=0;
```

```
  lambda=5;
```

```
  delta=0.01;
```

```
  output;
```

```

do i = 1 to 1000;
  t =t+delta;
  p = p+ rand('Poisson',delta*lambda);
  x=lambda*t;
  output;
end;
run;

```


\* Procedura rysowania krzywej zadanej w sposób dyskretny tzn. dane są wartości funkcji  $p$  i  $x$  w chwilach dyskretnych (tutaj co delta)\*

```


Symbol value=none interpol=sms line=1 width=2;
title"Poisson";
proc gplot data=poisson;
plot p*t x*t /overlay ;
run;

```

Obserwacja krzywej prowadzi do zrozumienia czym jest parametr lambda.



Rysunek dla lambda =5



Rysunek dla lambda=1

Umiejętność (wyznaczenia) symulacji trajektorii, pozwala nam myśleć, że znamy proces. Niestety obserwacja samych trajektorii może prowadzić do fałszywych wniosków. Wydaje się jakby proces Poissona mógł wykonać w danej chwili skok o dwie wartości, (np. tzn. jednoczesna rejestracja dwóch zdarzeń). Jest to związane z tym, że parametr  $\lambda$  może być relatywnie duży co wymusi prawdopodobieństwo skoków wyraźnie większych niż jeden. Niemniej skoki procesu Poissona są ZAWSZE równe jeden (w języku prawdopodobieństwa z prawdopodobieństwem jeden, prawie na pewno, prawie zawsze). Na rysunku skoki o więcej niż jeden oznaczają, że skoki nastąpiły wprawdzie w różnym czasie ale jednocześnie różnica pomiędzy tymi momentami skoku jest mniejsza niż  $\delta = 0.01$ .

Bardziej rozwinięty model to mieszany proces Poissona. To jest model w którym parametr  $\lambda$  jest sterowany niezależnie poprzez zmienną losową  $\Lambda$ . Wygląda to następująco, najpierw wybieramy  $\lambda$  wg. rozkładu zmiennej losowej  $\Lambda$ . Następnie przeprowadzamy całą analizę wybierając czasy pojawienia się rozszczeń.

Przeanalizujemy proces dla rozkładu Gamma wzór ogólny (symulacja)

$$f(x) = \frac{1}{\Gamma(a)} x^{a-1} e^{-x}.$$

```
data poissonmix;
t=0;
p=0;
delta=0.01;
lamb=rand('Gamma',16);
```

```

output;
do i = 1 to 1000;
  t =t+delta;
  p = p+ rand('Poisson',delta*lamb);
  output;
end;

run;

```

Inny model to intensywność zmieniająca się w czasie. Zakładamy, że proces Poissona oprócz warunków 1-2 spełnia zmodyfikowany warunek 3, czyli istnieje funkcja nieujemna, całkowalna taka, że przyrosty,  $N(t) - N(s)$ ,  $s < t$  mają rozkład Poissona z parametrem

$$\mu((s, t]) = \int_s^t \lambda(x) dx.$$

Jeśli  $\lambda(x) = x$  obserwujemy gwałtowny wzrost roszczeń, gdyż (pracujemy z krokiem  $delta = \delta = 0.01$ ) wówczas

$$\begin{aligned} \mu((t, t + \delta]) &= \int_s^{t+\delta} \lambda(x) dx = x^2/2|_t^{t+\delta} \\ &= ((t + \delta)^2 - t^2)/2 = \delta^2/2 + t\delta. \end{aligned}$$

Zatem przyrosty procesu  $N(t+\delta) - N(t)$  mają rozkład Poissona z parametrem  $\delta^2/2 + t\delta$ . Podstawiając  $c = \delta^2/2$  otrzymamy  $c + t * \delta$ .

```

data poissonmix2;

x=0;
t=0;
p=0;
delta=0.01;
c=delta**2/2;
output;

do i = 1 to 1000;
  t =t+delta;
  x=t**2/2;
  p = p+ rand('Poisson',c+t*delta);
  output;
end;

run;

```

Okazuje się, że twierdzenie o zmianie czasu (tzw. wprowadzenie czasu lokalnego) pozwala nam w inny sposób generować proces Poissona z funkcją intensywności  $\lambda(x)$ . Mianowicie niech  $N$  będzie jednorodnym procesem Poissona z parametrem  $\lambda = 1$ . Wprowadzamy lokalny czas

$$\mu(t) = \int_0^t \lambda(x) dx.$$

Wówczas proces

$$\tilde{N}(t) = N(\mu(t))$$

jest procesem Poissona o funkcji intensywności  $\lambda(x)$ .

**Uwagi.** Ten proces symulacji nie mówi nam gdzie występują skoki. Niemniej dobrze obrazuje ogólny stan procesu. Do generacji momentu skoków i tym samym procesu Poissona służy następująca procedura. Niech  $T_1, \dots, T_n$  będzie ciągiem zmiennych losowych o jednakowym rozkładzie wykładniczym z parametrem  $\lambda$ . Wówczas proces jednorodny Poissona (czyli z parametrem  $\lambda = 1$ ) możemy zdefiniować za pomocą formuły  $N(t) = 0$  gdy  $T_1 > t$  oraz jeśli  $T_1 \leq t$  wówczas

$$N(t) = \sup\{n : T_1 + T_2 + \dots + T_n \leq t\}.$$

Każdy "inny" proces Poissona otrzymamy wprowadzając lokalny czas.

## 2.2 Proces Wienera

### 2.2.1 Jednowymiarowy proces Wienera

Proces  $W(t)$ ,  $t \geq 0$  o własnościach

- a. Przyrosty niezależne, czyli dla dowolnych  $t_1 < t_2 < t_3 < t_4$ , zmienne losowe

$$W(t_2) - W(t_1), \quad W(t_4) - W(t_3)$$

są niezależne.

- b. Proces startuje z zera  $P(W(0) = 0) = 1$ .
- c. Przyrosty,  $W(t) - W(s)$ ,  $s < t$  mają rozkład normalny  $N(0, (t - s))$ , gdzie parametr  $t - s$  oznacza wariancję.

jest procesem Wienera.

Czasami się zakłada, że trajektorie procesu są ciągłe. Niemniej z twierdzenia Kołmogorowa wynika, że założenia 1-3 implikują ciągłość trajektorii. Co więcej są one Hölderowskie.

```

data wiener;
  x=0;
  z=0;
  delta=0.01;
  output;

  do i = 1 to 100;
    x =x+delta;
    z = z+ rand('normal')*sqrt(delta);
    output;
  end;
run;

```

Tworzona jest łamana z krokiem  $delta = \delta = 0.01$ . W powyższym programie łamana jest zdefiniowana za pomocą ciągu niezależnych zmiennych losowych `rand('normal')` nazwijmy je  $\xi_j$ . Cała procedura wyznaczenia i rysowania procesu może być zapisana wzorem

$$S_n(t) = \frac{\xi_1 + \dots + \xi_{[nt]}}{\sqrt{n}} + (nt - [nt]) \frac{\xi_{[nt]+1}}{\sqrt{n}}.$$

Tutaj  $n = 100 = \frac{1}{\delta}$ . Można udowodnić, że proces  $S_n$  zbiega słabo do procesu Wienera. Aby zrozumieć trochę zbieżność zobaczmy, że

$$S_n : \Omega \rightarrow C([0, t]),$$

gdzie  $C([0, T])$  oznacza zbiór funkcji ciągłych. Transportuje się miarę probabilistyczną na  $C([0, T])$  i miara ta zbiega do miary Wienerowskiej.

### 2.2.2 Konstrukcja Ciesielskiego procesu Wienera

Skonstruujemy proces na odcinku  $[0, 1]$ . Ciesielski wykorzystał tutaj bazę Haara. ( To ciąg funkcji, które tworzą bazę w przestrzeni Hilberta  $L^2([0, 1])$ . W oparciu o tę bazę przesyła się w sieci obrazy.) Definicja tych funkcji jest następująca.

Funkcja  $h_0$  jest funkcją stałą równa jeden,

$$h_0(t) = 1, \quad t \in [0, 1].$$

Dla danej liczby  $n$  istnieje jej diadyczne przedstawienie, czyli istnieją liczby całkowite  $k, j$  takie, że

$$n = 2^k + j, \quad 0 \leq j < 2^k.$$

Na przykład

$$13 = 2^3 + 5, \quad 20 = 2^4 + 4.$$

Funkcję Haara o indeksie  $n$  wyznaczamy ze wzoru

$$h_n(t) = \begin{cases} 2^{\frac{k}{2}} & \text{dla } (2j)2^{-(k+1)} \leq t < (2j+1)2^{-(k+1)} \\ -2^{\frac{k}{2}} & \text{dla } (2j+1)2^{-(k+1)} \leq t < (2j+2)2^{-(k+1)} \\ 0 & \text{dla } \text{poza} \end{cases}$$

Niech  $\xi_j$  będzie ciągiem niezależnych zmiennych losowych o rozkładzie normalnym  $N(0, 1)$ . Wówczas proces

$$W(t) = \sum_{n=0}^{\infty} \xi_n \int_0^t h_n(s) ds$$

jest procesem Wienera.

Konstrukcja ta uogólniona później została przez Ito Nisio. Pozwala ona (w przeciwieństwie do procedury wcześniejszej) „przybliżyć” obraz trajektorii procesu Wienera. Skonstruujemy algorytm oparty o powyższy wzór. Mianowicie zauważmy, że funkcje

$$\varphi_n(t) := \int_0^t h_n(s) ds$$

są kawałkami liniowe. Ponadto funkcja  $\varphi_n$  dla  $n = 2^k + j$  zeruje się w punktach diadycznych niższej generacji, czyli dla wszystkich  $0 \leq m \leq 2^{k+1}$

$$\varphi_n\left(\frac{m}{2^{k+1}}\right) = 0.$$

Założmy że dany jest proces

$$\psi_K(t) := \sum_{n=0}^{2^K-1} \xi_n \varphi_n(t).$$

Ponieważ jest on kawałkami liniowy to oznacza, że znamy jego wartości w punktach diadycznych

$$t = \frac{i}{2^K}, \quad 0 \leq i \leq 2^K.$$

Zatem aby wyznaczyć

$$\psi_{K+1}(t) = \sum_{n=0}^{2^{K+1}-1} \xi_n \varphi_n(t)$$

$$= \psi_K(t) + \sum_{n=2^K}^{2^{K+1}-1} \xi_n \varphi_n(t),$$

wystarczy wyznaczyć funkcję  $\psi_{K+1}$  w punktach diadycznych kolejnej generacji, czyli w punktach

$$t = \frac{2i+1}{2^{K+1}}, \quad 0 \leq i < 2^K,$$

gdyż w punktach  $t = \frac{i}{2^K}$ ,  $0 \leq i \leq 2^K$

$$\psi_K(t) = \psi_{K+1}(t).$$

Z definicji funkcji  $\varphi_n$  wynika, że dla

$$t = \frac{2i+1}{2^{K+1}}, \quad 0 \leq i < 2^K,$$

$$\psi_{K+1}\left(\frac{2i+1}{2^{K+1}}\right) = \frac{\psi_K\left(\frac{i}{2^K}\right) + \psi_K\left(\frac{i+1}{2^K}\right)}{2} + \left(\frac{1}{\sqrt{2}}\right)^{K+1} \xi,$$

gdzie  $\xi$  t zmienna losowa o rozkładzie  $N(0,1)$  niezależna od wcześniej wygenerowanych zmiennych losowych.

```
data wiener_ciesielski;

K=10;
N=1023; *N=2**K-1 ilosc funkcji*
array W[1025] W1-W1025; *N+2*

W[1]=0;
W[2**K+1]=rand('normal');

do j= 1 to K;
  do i=0 to 2**(j-1)-1;
    r=((2*i+1)*(2**(K-j)))+1;
    s1=(i)*(2**(K-j+1))+1;
    s2=(i+1)*(2**(K-j+1))+1;
    W[r]=(W[s1]+W[s2])/2+(1/sqrt(2))**(j+1)*rand('normal');
  end;
end;

do j=0 to N+1;
  x=W[j+1];
```

```

    t=j/2**K;
    output;
end;
run;

Symbol value=none interpol=sms line=1 width=2;
title"Wiener";
proc gplot data=wiener_ciesielski;
plot x*t;
run;

```

### 2.2.3 Dwuwymiarowy rozkład normalny

Zanim przejdziemy do konstrukcji dwóch procesów skorelowanych Wienera skonstruujemy dwuwymiarowy rozkład normalny  $(X, Y)$ . Dwuwymiarowy rozkład normalny charakteryzuje gęstość dwuwymiarowa

$$f(x, y) = \frac{1}{2\pi\sqrt{\det\Sigma}} e^{-\langle Q(x-m), x-m \rangle},$$

gdzie  $\langle x, y \rangle$  oznacza iloczyn skalarny w  $R^2$ . Przypomnijmy dla wektorów  $x, y \in R^2$

$$\langle x, y \rangle = x_1y_1 + x_2y_2, \quad x = (x_1, x_2), y = (y_1, y_2).$$

Dwuwymiarowy rozkład normalny ma pięć parametrów co zapisujemy  $N(m, Q)$ . Wyjaśnimy sens parametrów. Parametr  $m = (m_1, m_2)$  jest wektorem wartości oczekiwanych, czyli

$$EX = m_1, \quad EY = m_2.$$

Macierz  $Q$  jest odwrotna do macierzy kowariancji  $\Sigma$ , czyli

$$Q\Sigma = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}.$$

Macierz kowariancji

$$\Sigma = \begin{bmatrix} \text{Var}(X) & \text{Cov}(X, Y) \\ \text{Cov}(X, Y) & \text{Var}(Y) \end{bmatrix}.$$

Pary punktów o rozkładzie normalnym  $N(m, Q)$  generujemy korzystając z twierdzenia, że każdy wektor normalny otrzymywany jest za pomocą wektora

niezależnych zmiennych losowych  $(Z_1, Z_2)$  o rozkładzie  $N(0, 1)$ . Czyli istnieje macierz  $A$  taka, że

$$\begin{bmatrix} X \\ Y \end{bmatrix} = A \begin{bmatrix} Z_1 \\ Z_2 \end{bmatrix} + m.$$

Teraz jesteśmy gotowi na generację próbki o dwuwymiarowym rozkładzie normalnym

```
data twonormal;
```

```
do i=1 to 5000;
  z1=rand('normal');
  z2=rand('normal');
  x1=1+ z1-z2;
  x2=2+2*z2;
  output;
end;
```

```
run;
```

Proces tworzenia dwuwymiarowej gęstości zarówno wykres konturowy i powierzchniowy oraz wygenerowane pary punktów daje procedura:

```
ods graphics on;
```

```
proc kde data=twonormal;
  bivar x1 x2 / plots=all;
run;
```

```
ods graphics off;
```

#### 2.2.4 Dwuwymiarowy proces Wienera

Tworzymy dwie niezależne kopie procesu Wienera  $z1 = z_1(t)$  oraz  $z2 = z_2(t)$  oraz proces do nich skorelowany  $b(t)$ , czyli

$$b(t) = \eta z_1(t) + \rho z_2(t),$$

gdzie parametry są stałe i spełniają zależność

$$\eta^2 + \rho^2 = 1.$$

```

data wiener_skorelowany;
  x=0;
  z1=0;
  z2=0;
  b=0;
  delta=0.01;
  output;
  rho=0.8;
  eta=sqrt(1-rho**2);
  do i = 1 to 100;
    x =x+delta;
    z1 = z1+rand('normal')*sqrt(delta);
    z2= z2+rand('normal')*sqrt(delta);
    b=eta*z1+rho*z2;
    output;
  end;
run;

```

### 3 Równania stochastyczne

#### 3.1 Proces Ito formuła Ito

Proces Ito jest to proces postaci

$$dX_t = a_t dt + b_t dW_t. \quad (1)$$

Prawidłowo powinno się używać postaci całkowej

$$X_t = X_0 + \int_0^t a_s ds + \int_0^t b_s dW_s.$$

Założenia, które spełniają procesy  $a_s$  i  $b_s$  są tak dobrane aby powyższe całki miały sens. Druga całka to całka Ito. Istotne jest, że możemy używać następującej procedury

$$\Delta X_t = a_t \Delta t + b_t \Delta W_t,$$

do otrzymania procesu  $X_t$ , gdzie  $t \in [0, T]$ . Rozważmy rozbicie odcinka  $[0, 1]$ , czyli wprowadzamy czas dyskretny  $0 = t_0 < t_1 < t_2 < \dots < t_n = T$ , np. ustalając jednakowe odstępów czasowe

$$t_{j+1} - t_j = \frac{T}{\delta}.$$

Wówczas rozpatrujemy rekurencyjne równanie

$$X_{t_{j+1}} - X_{t_j} = a_{t_j}(t_{j+1} - t_j) + b_{t_j}(W_{t_{j+1}} - W_{t_j}).$$

Wychodząc od znanej  $X_0$  otrzymamy dyskretny proces (zależny od rozbicia), który jest zbieżny do szukanego procesu  $X_t$ .

Ważny fakt brzmi, że jeśli złożymy proces Ito  $X_t$  z funkcją dwóch argumentów  $F(t, x)$  dostatecznie gładką, w taki sposób, że otrzymamy nowy proces

$$Y_t = F(t, X_t),$$

to otrzymam dalej proces Ito o dynamice

$$dF(X_t, t) = \frac{\partial F}{\partial t}(t, X_t)dt + \frac{\partial F}{\partial x}(t, X_t)dX_t + \frac{1}{2} \frac{\partial^2 F}{\partial x^2}(t, X_t)b_t^2 dt.$$

**Zadanie** załóżmy, że dany jest proces

$$dX_t = adt + bdW_t, \tag{2}$$

gdzie  $a, b$  to pewne stałe. Wyznaczyć dynamikę  $e^{X_t}$ .

**Uwagi.** Zanim rozwiążemy problem znalezienia dynamiki zapiszmy (2) w postaci całkowej

$$X_t = X_0 + \int_0^t ads + \int_0^t bdW_s.$$

Pierwsza całka jest zwyczajną całką oznaczoną. Dla drugiej korzystając z elementarnych własności całki Ito otrzymamy, że proces

$$X_t = X_0 + at + bW_t,$$

czyli jest to przeskalowany proces Wienera z tzw. dryftem.

```
data wiener_dryft;
  t=0;
  z=0;
  delta=0.01;
  a=2;
  b=3;
  output;
  do i = 1 to 1000;
    t =t+delta;
    z = z+ b*rand('normal')*sqrt(delta)+a*delta;
    output;
  end;
run;
```

**Rozwiązanie.** Aby skorzystać z formuły Ito zauważmy, że

$$F(x) = e^x.$$

Funkcja zależy tylko od jednego parametru zatem pochodna po czasie znika. Zauważmy, że

$$F'(x) = e^x, \quad F''(x) = e^x.$$

Stąd po wstawieniu do formuły Ito dla procesu  $Y_t = e^{X_t}$  otrzymamy

$$dY_t = e^{X_t}dX_t + \frac{1}{2}e^{X_t}b^2dt.$$

Po wstawieniu  $Y_t = e^{X_t}$  oraz używając (2) otrzymamy dynamikę w postaci równania stochastycznego

$$dY_t = Y_t(adt + bdW_t) + \frac{1}{2}Y_t b^2 dt = Y_t(a + \frac{1}{2}b^2)dt + Y_t b dW_t.$$

Oczywiście proces

$$Y_t = Y_0 e^{at+bW_t}$$

jest rozwiązaniem powyższego równania.

### 3.2 Równania stochastyczne

Równani różniczkowe stochastyczne zapisujemy w postaci

$$dX_t = f(X_t, t)dt + g(X_t, t)dW_t. \quad (3)$$

Przypominamy, że matematycznie jest to równanie całkowe typu

$$X_t = X_0 + \int_0^t f(X_s, s)ds + \int_0^t g(X_s, s)dW_s$$

ale ze względów historycznych używa się części równania (3). Przy założeniach, że funkcje są typu Lipschitza istnieje rozwiązanie równania (3). Do sprawdzania, czy dany proces spełnia równanie stochastyczne służy formuła Ito. (Jest to najczęściej wykorzystywana metoda, czyli podawane jest rozwiązanie i następnie sprawdza się, że spełnia ono równanie (3)).

### 3.3 Model Samuelsona - Geometryczny proces Wienera

W modelu Blacka-Scholesa zakładamy, że dynamika zmian cen akcji (aktywu) dana jest wzorem

$$dS_t = S_t \mu dt + S_t \sigma dW_t.$$

Parametry:  $\mu$  oznacza oczekiwany wzrost cen akcji, zaś  $\sigma$  odchylenie standardowe zmian cen akcji.

**Zadanie.** Korzystając z formuły Ito sprawdzić, że proces

$$S_t = S_0 e^{(\mu - \frac{1}{2}\sigma^2)t + \sigma W_t} \quad (4)$$

jest rozwiązaniem powyższego równania.

**Uwagi.** Z postaci (4) wynika, że ceny akcji mają rozkład log-normalny. Ta obserwacja została pierwszy raz zrobiona przez Samuelsona. Powyższą obserwację możemy potwierdzić generując rozwiązanie. Zakładamy, że cena początkowa  $S_0 = 5$ ,  $\mu = 0.08$ ,  $\sigma = 0.4$ . Obserwujemy  $S_t/S_0$  dla  $t = 1$ .

```
data black;
mi=0.08;
sigma=0.4;
delta=0.01;

do k=1 to 400; *Puszczamy 400 razy trajektorie*
  S=5;
  do j = 1 to 100;
    y=rand('normal');
    S=S*(1+mi*delta+sigma*y*sqrt(delta));
  end;
  c=S/5;
output;
end;
run;

title 'Distribution lognormal';
ods select ParameterEstimates
      GoodnessOfFit FitQuantiles MyHist;
proc univariate data=black;
var c;
histogram / midpoints=0.1 to 3 by 0.2
```

```

lognormal
gamma
vaxis    = axis1
name     = 'MyHist';
inset n mean(5.3) std='Std Dev'(5.3) skewness(5.3)
      / pos = ne header = 'Summary Statistics';
axis1 label=(a=90 r=0);
run;

```

**Rozwiązanie.** Korzystamy z formuły Ito dla funkcji  $F(x) = e^x$ .

### 3.4 Proces Ornstein - Uhlenbecka

Proces postaci

$$dX_t = \theta((\mu - X_t)dt + \sigma dW_t) \quad (5)$$

nazywamy procesem OrnsteinUhlenbecka. Tym razem nie zastosujemy bezpośredniej metody rozwiązania równania. Mianowicie rozważmy proces

$$f(X_t, t) = X_t e^{\theta t},$$

gdzie  $X_t$  jest rozwiązaniem (5). Oczywiście jest to proces Ito. Wówczas z foruły Ito wynika, że

$$df(X_t, t) = \theta X_t e^{\theta t} dt + e^{\theta t} dX_t.$$

Wstawiając z (5)  $dX_t$  otrzymamy

$$\begin{aligned} df(X_t, t) &= \theta X_t e^{\theta t} dt + e^{\theta t}(\theta((\mu - X_t)dt + \sigma dW_t)) \\ &= \theta X_t e^{\theta t} dt + e^{\theta t}\theta\mu dt - e^{\theta t}\theta X_t dt + e^{\theta t}\sigma dW_t \\ &= e^{\theta t}\theta\mu dt + e^{\theta t}\sigma dW_t \end{aligned}$$

Stąd

$$X_t e^{\theta t} = X_0 + \int_0^t e^{\theta s}\theta\mu ds + \int_0^t e^{\theta s}\sigma dW_s.$$

Zatem rozwiązując całki

$$X_t e^{\theta t} = X_0 + \mu(e^{\theta t} - 1) + \int_0^t e^{\theta s}\sigma dW_s.$$

Stąd rozwiązaniem równania (5) jest proces

$$X_t = X_0 e^{-\theta t} + \mu(1 - e^{-\theta t}) + \int_0^t e^{\theta(s-t)}\sigma dW_s.$$

Ustalając  $t = T$  rozwiązanie  $X_T$  jest zatem zmienną losową gaussowską. Ponadto jeśli całkę oznaczmy przez  $U$ ,

$$U = \int_0^T e^{\theta(s-T)} \sigma dW_s,$$

to jest ona zmienną losową gaussowską o wartości oczekiwanej  $EU = 0$ , zaś wariancja

$$VarU = (\sigma e^{-\theta T})^2 \int_0^T e^{2\theta s} ds.$$

Stąd

$$VarU = \sigma^2 e^{-2\theta T} \frac{e^{2\theta T} - 1}{2\theta} = \sigma^2 \frac{1 - e^{-2\theta T}}{2\theta}$$

Jeśli  $X_0$  jest stała np. równą  $x_0$ , to  $X_T$  jest zmienną losową o rozkładzie normalnym z średnią

$$EX_T = X_0 e^{-\theta T} + \mu(1 - e^{-\theta T}).$$

zaś

$$VarX_T = VarU.$$

Przeanalizujemy ten przypadek również od strony symulacji

\*Warunki początkowe\*

```
data ornstein;
theta=0.08;
sigma=0.4;
mi=0.01;
krok=0.01;
X=5;
```

Z tymi warunkami początkowymi ( $T = 1$ )

$$EX_1 = 5 * e^{-0.08} + 0.01(1 - e^{-0.08}) = 4.61$$

$$VarX_1 = (0.4)^2 \frac{1 - e^{-2*0.08}}{2 * 0.08} = 0.38.$$

Stąd odchylenie standardowe wynosi

$$\sigma(X_1) = 0.38.$$

Generowanie trajektorii (ścieżek) procesu z krokiem  $krok = 0.01$ , które obserwujemy po dojściu do "ekranu" bariery w chwili  $T = 1$ . Generowanych jest 1000 trajektorii.

```

do k=1 to 1000;
  x=5;
  do j = 1 to 100;
    y=rand('normal');
    x=x+theta*(mi-x)*krok+sigma*y*sqrt(krok);
  end;
  X=x;
  output;
end;
run;

```

### 3.5 Model Hestona

Cena akcji w modelu Hestona spełnia równanie

$$dS_t = \mu S_t dt + \sigma_t S_t dZ_t^1, \quad (6)$$

gdzie  $\mu$  oczekiwana stopa zwrotu. Odchylenie standardowe  $\sigma_t$  jest procesem, który spełnia równanie Ornsteina-Uhlenbecka

$$d\sigma_t = -\beta\sigma_t dt + \delta dZ_t^2. \quad (7)$$

Para  $(Z_t^1, Z_t^2)$  jest dwuwymiarowym procesem Wienera o korelacji  $\rho$ , czyli  $Cov(Z_t^1, Z_t^2) = \rho t$ .

Ten układ równań (6) i (7) jest przedstawiany często w postaci alternatywnej. Zdefiniujmy nowy proces

$$\nu_t = \sigma_t^2.$$

Wówczas z formuły Ito

$$d\nu_t = 2\sigma_t d\sigma_t + \frac{1}{2}2\delta^2 dt.$$

Stąd wstawiając równanie (7) oraz podstawiając  $\sigma_t = \sqrt{\nu_t}$  otrzymamy

$$d\nu_t = 2\sqrt{\nu_t}(-\beta\sqrt{\nu_t}dt + \delta dZ_t^2) + \frac{1}{2}2\delta^2 dt.$$

Zatem

$$d\nu_t = 2\delta\sqrt{\nu_t}dZ_t^2 + (-\beta\nu_t + \delta^2)dt$$

Jest to równanie typu CIR (Cox, Ingresoll, Ross), którym modeluje się stopy procentowe.

Zatem równoważny układ równań z (6) i (7) jest następujący

$$\begin{cases} dS_t = \mu S_t dt + \sqrt{\nu_t} S_t dZ_t^1 \\ d\nu_t = 2(\delta^2 - \beta \nu_t) dt + \delta \sqrt{\nu_t} dZ_t^2. \end{cases}$$

Zakładamy, że znane są warunki początkowe: cena akcji  $S_0 = 5$ , wyjściowa wariancja  $\nu_0 = 0.08$ . Ponadto znany jest współczynnik korelacji  $\rho = 0.9$  pomiędzy procesami  $Z^1$  a  $Z^2$ . Znane są również pozostałe parametry modelu czyli:  $\mu = 0.05$ ,  $\beta = 0.02$  oraz  $\delta = 0.05$ .

data Heston;

\* warunki początkowe\*

```
x=0;
w=0;
z1=0;
z2=0;
S=5;
ni=0.08;
output;

rho=0.9;
eta=sqrt(1-rho**2);
mi=0.05;
krok=0.01;* krok czasowy*
delta=0.05;
beta=0.02;
```

\* procedura generowania procesów Wienera Z1 i Z2 oraz procesów ni i S\*

```
do i = 1 to 1000;
  y1 = rand('normal');
  y2 = rand('normal');
  x =x+krok;
  *proces z2 w chwili x*
  z2u=rho*z1+eta*w;
  *procesy z1 i w niezależne*
  z1 = z1+ y1*sqrt(krok);
  w= w+y2*sqrt(krok);
  *proces z2 w chwili x+krok*
```

```

z2=rho*z1+eta*w;
ni=ni+2*(delta**2-beta*ni)*krok+delta*sqrt(ni)*(z2-z2u);
S=S*(1+mi*krok+sqrt(ni)*y1*sqrt(krok));
output;
end;
run;

```

```

Symbol value=none interpol=sms line=1 width=2;
legend1 order=( 'z1' 'z2');
title"Wiener skorelowany";
proc gplot data=Heston;
plot z1*x z2*x/overlay legend=legend1 ;
run;

```

Oczywiście możemy znaleźć rozwiązania dokładne procesu. Mianowicie równanie (7) zadaje proces Ornsteina Uhlenbecka zatem

$$\sigma_t = \sigma_0 e^{-\beta t} + \delta \int_0^t e^{\beta(s-t)} dZ_s^2$$

Stąd ponieważ model cen akcji w modelu Hestona nie różni się istotnie od modelu Blacka-Scholesa, to

$$S_t = S_0 \exp \left\{ \left( \mu - \frac{\sigma_t^2}{2} \right) t + \sigma_t Z_t^1 \right\}.$$

Wiemy, że  $\sigma_t$  to proces gaussowski zatem  $S_t$  to zmienna losowa, taka że  $\ln S_t$  ma rozkład będący iloczynem współrzędnych wektora gaussowskiego.

### 3.6 Uogólniony model Hestona

Możemy w naszych rozważaniach uwzględnić również zmieniającą się w czasie stopę procentową, tzw. stopę procentową dla lokaty krótkoterminowej. Wówczas rozważa się układ równań (L. Grzelak, C. W. Oosterlee)

$$\begin{cases} \frac{dS_t}{S_t} = r_t dt + \sqrt{\nu_t} dZ_t^1 \\ d\nu_t = 2\beta \left( \frac{\delta^2}{\beta} - \nu_t \right) dt + \delta \sqrt{\nu_t} dZ_t^2 \\ dr_t = \lambda(\theta - r_t) dt + \eta \sqrt{r_t} dZ_t^3. \end{cases}$$

### 3.7 Model SABR

Model zadany jest układem równań

$$dF_t = \alpha_t F_t^\beta dZ_t^1$$

oraz

$$d\alpha_t = \nu \alpha_t dZ_t^2$$

Procesy Wienera  $Z_t^1$  i  $Z_t^2$  są skorelowane z parametrem  $\rho$ . Dane są warunki początkowe

$$F(0) = f, \quad \alpha(0) = \alpha.$$

Nazwa modelu nawiązuje do nazwy parametrów w modelu tj. stochastic  $(\alpha, \beta, \rho)$ . Zauważmy, że nieznanym jest również warunek początkowy  $f$ . Rozwiązania analityczne są znane (jak dotąd) tylko dla  $\beta = 0$  oraz  $\beta = 1$ .

### 3.8 Miara martyngałowa-kilka uwag o dynamice cen akcji

Modele rozpatrywane były względem tzw. miary rzeczywistej, czyli względem "rzeczywistych" możliwych ścieżek procesu cen akcji (wyjątek stanowią dwa ostatnie modele). Istotą wyceny jest taki model by był on równoważny z brakiem arbitrażu. Taki model zasadniczo prowadzi do modelu gdzie zdyskontowane ceny akcji  $\tilde{S}_t := e^{-rt} S_t$  są martyngałem,  $r$  to stopa procentowa dla kapitalizacji ciągłej. Zastosujmy formułę Ito dla funkcji  $F(t, x) = e^{-rt} x$  oraz procesu  $S_t$ . Wówczas

$$d\tilde{S}_t = dF(t, S_t) = -r e^{-rt} S_t dt + e^{-rt} dS_t.$$

Wstawiając dynamikę cen akcji

$$dS_t = S_t \mu dt + S_t \sigma dW_t$$

otrzymamy

$$d\tilde{S}_t = dF(t, S_t) = -r e^{-rt} S_t dt + e^{-rt} (S_t \mu dt + S_t \sigma dW_t).$$

Stąd

$$d\tilde{S}_t = dF(t, S_t) = e^{-rt} S_t ((\mu - r) dt + \sigma dW_t) = \tilde{S}_t \sigma \left( \frac{\mu - r}{\sigma} dt + dW_t \right).$$

Zatem jeśli proces dryftu procesu Wienera

$$dB_t = \frac{\mu - r}{\sigma} dt + dW_t.$$

z innego punktu widzenia (innej miary) znów będzie procesem Wienera, to zdyskontowany proces  $\tilde{S}_t$  będzie martyngałem. Wynika to ze wzoru

$$d\tilde{S}_t = \tilde{S}_t \sigma dB_t.$$

Wracając do dynamiki samego procesu  $S_t$  otrzymamy w modelu Blacka Scholesa równanie

$$dS_t = rS_t dt + S_t \sigma dB_t.$$

## 4 Elementy statystyki

Do tej pory parametry procesów były zadane. W rzeczywistości są one szacowane z danych historycznych. Wrócimy do modelu Blacka-Scholesa

$$dS_t = S_t \mu dt + S_t \sigma dW_t.$$

Model ten ma dwa parametry  $\mu$  oraz  $\sigma$ . Z danych historycznych możemy oszacować parametr  $\sigma$  a następnie  $\mu$ . Naiwny estymator wariancji

$$\widehat{\sigma}^2 = \frac{1}{n} \sum_{i=1}^n \frac{(\ln S_{t_i} - \ln S_{t_{i-1}})^2}{t_i - t_{i-1}} - \frac{1}{n} \frac{(\ln S_{t_n} - \ln S_{t_0})^2}{t_n - t_0},$$

gdzie korzystamy z czasu dyskretnego  $0 = t_0 < t_1 < t_2 < \dots < t_n = T$ . Korzystając z rozwiązania (4)

$$\ln \frac{S_t}{S_s} = \left(\mu - \frac{1}{2}\sigma^2\right)(t - s) + \sigma(W_t - W_s)$$

otrzymamy

$$\begin{aligned} \widehat{\sigma}^2 &= \\ & \frac{1}{n} \sum_{i=1}^n \left( \frac{(\mu - \frac{1}{2}\sigma^2)(t_i - t_{i-1}) + \sigma(W_{t_i} - W_{t_{i-1}})}{\sqrt{t_i - t_{i-1}}} \right)^2 - \frac{1}{n} \left( \frac{(\mu - \frac{1}{2}\sigma^2)T + \sigma W_T}{\sqrt{T}} \right)^2. \\ &= \frac{1}{n} \sum_{i=1}^n \left( (\mu - \frac{1}{2}\sigma^2)\sqrt{t_i - t_{i-1}} + \sigma \frac{(W_{t_i} - W_{t_{i-1}})}{\sqrt{t_i - t_{i-1}}} \right)^2 - \\ & \quad \frac{1}{n} \left( (\mu - \frac{1}{2}\sigma^2)\sqrt{T} + \sigma \frac{W_T}{\sqrt{T}} \right)^2. \end{aligned}$$

Zatem wprowadzając zmienne losowe niezależne  $\xi_i$  o rozkładzie  $N(0, 1)$  otrzymamy

$$\widehat{\sigma}^2 = \frac{1}{n} \sum_{i=1}^n \left( (\mu - \frac{1}{2}\sigma^2)\sqrt{t_i - t_{i-1}} + \sigma \xi_i \right)^2 - \frac{1}{n} \left( (\mu - \frac{1}{2}\sigma^2)\sqrt{T} + \sigma \xi \right)^2.$$

Stąd korzystając z  $EX^2 = VarX + (EX)^2$  otrzymamy

$$E\widehat{\sigma}^2 = \sigma^2 + \frac{1}{n} \sum_{i=1}^n (\mu - \frac{1}{2}\sigma^2)^2 (t_i - t_{i-1}) - \frac{\sigma^2}{n} - (\mu - \frac{1}{2}\sigma^2)^2 \frac{T}{n} = \frac{n-1}{n} \sigma^2$$

Przedstawimy metodę estymacji wariancji korzystającą ze wzorów Blacka-Scholesa dla wyceny opcji. Jak estymować  $\mu$ . Niech

$$L = \frac{1}{T} \sum_{i=1}^n (\ln S_{t_i} - \ln S_{t_{i-1}}).$$

Korzystając z (4)

$$L = \frac{1}{T} \sum_{i=1}^n \left( (\mu - \frac{1}{2}\sigma^2)(t_i - t_{i-1}) + \sigma(W_{t_i} - W_{t_{i-1}}) \right) = (\mu - \frac{1}{2}\sigma^2) + \frac{\sigma W_T}{T}.$$

Ponieważ  $EL = \mu - \frac{1}{2}\sigma^2$  zatem

$$\hat{\mu} = L - \frac{n}{2(n-1)} \widehat{\sigma}^2.$$